

Appendix E

OpenBUGS code for the BLCA model from Chapter 4

Te tohu OpenBUGS mō te tauira BLCA i te Wāhanga 4

```

model{
#Multinomial Model for the Data

#x1 is the test results in high prevalence population, the 4 combinations follow a multinomial
#distribution
x1[1:2,1:2] ~ dmulti(p1[1:2,1:2], n1)
x2[1:2,1:2] ~ dmulti(p2[1:2,1:2], n2) #Observed prevalence

p1[1,1] <- pi1*Se1*Se2+(1-pi1)*(1-Sp1)*(1-Sp2) #both tests are positive
p1[1,2] <- pi1*Se1*(1-Se2)+(1-pi1)*(1-Sp1)*Sp2 #visual assessment positive, soil culture negative
p1[2,1] <- pi1*(1-Se1)*Se2+(1-pi1)*Sp1*(1-Sp2) #visual assessment negative, soil culture positive
p1[2,2] <- pi1*(1-Se1)*(1-Se2)+(1-pi1)*Sp1*Sp2 #both tests are negative

p2[1,1] <- pi2*Se1*Se2+(1-pi2)*(1-Sp1)*(1-Sp2)
p2[1,2] <- pi2*Se1*(1-Se2)+(1-pi2)*(1-Sp1)*Sp2
p2[2,1] <- pi2*(1-Se1)*Se2+(1-pi2)*Sp1*(1-Sp2)
p2[2,2] <- pi2*(1-Se1)*(1-Se2)+(1-pi2)*Sp1*Sp2

# Priors
pi1 ~ dbeta(14.59, 14.59) # High, "40-50, some at 80" 95% sure >35% and most likely at 50%
pi2 ~ dbeta(53.88, 1270.24) # low less than 5%, most likely set at 4%
Se1 ~ dbeta(4.53, 5.32) #for aerial inspection min=26, most likely = 45, max=66
Sp1 ~ dbeta(91.80, 7.83) #min = 89, most likely = 93, max = 97

```

```
Se2 ~ dbeta(23.12, 9.18)    # soil culture min = 65, most likely = 73, max = 89
Sp2 ~ dbeta(60.04, 6.13)   # min = 86, most likely = 92, max = 95
}
```

```
#Data
```

```
list(n1=189, n2=572)
```

```
#n1 is for high prevalence
```

```
# test 1 (visual) in rows, test 2 (soil culture) in columns
```

```
x1[,1] x1[,2] x2[,1] x2[,2]
```

```
22     26     8     73
```

```
35    106    11    480
```

```
END
```

```
#Initial values for the 3 chains
```

```
list(Se1=0.8, Sp1=0.2, Se2=0.2, Sp2=0.2, pi1=0.1, pi2=0.8)
```

```
list(Se1=0.45, Sp1=0.93, Se2=0.73, Sp2=0.92, pi1=0.61, pi2=0.17)
```

```
list(Se1=0.2, Sp1=0.98, Se2=0.9, Sp2=0.99, pi1=0.9, pi2=0.05)
```